How to use K3nCones

Nicolas Addington

Version 0.2
September 15, 2021

This document describes a Macaulay2 package to compute movable cones and their nef chambers for varieties of $K3^{[n]}$-type and Picard rank 2. Eventually I'll write proper documentation, and a paper explaining the algorithm.

I've done my best to test the program thoroughly, but it may still have major bugs. Please don't rely on it for your research while the version number is < 1. Send bug reports or suggested improvements to adding@uoregon.edu.

Installation

Download https://pages.uoregon.edu/adding/K3nCones.m2 and put the file somewhere on your computer. Open Macaulay2 and run

```
i1 : loadPackage("K3nCones", FileName => "path/to/where/you/put/it/K3nCones.m2")
o1 = K3nCones
o1 : Package
```

Hilbert schemes of points on K3 surfaces

Let's start with the Hilbert scheme of 7 points on a (very general) K3 surface of degree 2:

```
i2 : L = Hilb^7(2)
o2 : a lattice of K3^[7] type
o2 : K3nLattice
```

You can inspect some of its properties:

```
i3 : L.mukaiPairing
o3 = | 0  0 -1 |
     | 0  2 0  |
     | -1 0 0  |
               3        3
o3 : Matrix ZZ  <--- ZZ
```

```
i4 : L.mukaiVector
o4 = | 1  |
     | 0  |
     | -6 |
       3
o4 : ZZ

i5 : L.ampleClass
o5 = | 1  |
     | -5 |
     | 6  |
       3
o5 : ZZ
```

There are functions to take dot products and squares using the Mukai pairing:

```
i6 : L.dot(vector{1,2,3}, vector{4,5,6})
o6 = 2

i7 : L.square(vector{7,8,9})
o7 = 2
```

You can ask for the positive cone, or the movable cone, or the nef cone:

```
i8 : positiveCone(L)
o8 = {| -.0548022 |, | .160065  |}
      | -.134237  |  | -.392078 |
      | -.328813  |  | .960392  |
o8 : List

i9 : movableCone(L)
Searching v.w = 0...
Searching v.w = 1...
Searching v.w = 2...
o9 = {| 0  |, | 2  |}
      | -1 |  | -5 |
      | 0  |  | 12 |
o9 : List

i10 : nefCone(L)
Searching v.w = 3...
Searching v.w = 4...
Searching v.w = 5...
Searching v.w = 6...
o10 = {| 0  |, | 1  |}
       | -1 |  | -4 |
       | 0  |  | 6  |
o10 : List
```

Or if you prefer the answers in the basis $\tilde{H}$, B from Bayer and Macrì's papers:

```
i11 : nefCone(L, Basis => "HtildeB")
o11 = {| 1 |, | 4  |}
       | 0 |  | -1 |
o11 : List
```

So the walls of the nef cone are spanned by $\tilde{H}$ and $4\tilde{H} - B$, or you might prefer to say $\tilde{H} - \frac{1}{4}B$.

You can ask for *all* the nef chambers of the movable cone, which is probably the best way to use the program:

```
i12 : nefChambers(L, Basis => "HtildeB")
o12 = {| 1 |, | 4  |, | 7  |, | 3  |, | 11 |, | 8  |, | 5  |}
       | 0 |  | -1 |  | -2 |  | -1 |  | -4 |  | -3 |  | -2 |
o12 : List
```

This agrees with Bayer and Macrì's [3, Example 13.5], although it omits the "fake wall" $17\tilde{H} - 6B$ which they seem to have included just to trick us.

Moduli spaces of higher-rank sheaves

Let's take the moduli space of stable sheaves with Mukai vector $(2, h, 2)$ on a (very general) K3 surface of degree 10. This is Mukai's first example of a moduli space that's not birational to any Hilbert scheme.

```
i13 : L = moduliSpace(vector{2,1,2}, 10)
o13 : a lattice of K3^[2] type
o13 : K3nLattice
```

The movable cone only has one chamber – that is, the nef and movable cones coincide.

```
i14 : nefChambers(L)
Searching v.w = 0...
Searching v.w = 1...
o14 = {| 0  |, | 5 |}
       | -1 |  | 1 |
       | -5 |  | 0 |
o14 : List
```

So there's only one birational model.

I can't do moduli spaces of sheaves of rank 0 yet, because I haven't thought about how to choose the ample class; please email me if you know how to do it. I haven't bothered with moduli spaces of twisted sheaves, but I'll do it if someone twists my arm.

Lines on a cubic fourfold

Let's take the variety of lines on a (very general) cubic fourfold containing a plane. In Hassett's framework, that's a special cubic of discriminant 8:

```
i15 : L = fanoOfLines(8)
o15 : a lattice of K3^[2] type
o15 : K3nLattice
```

The basis for $K_{\mathrm{num}}(\mathcal{A}_X)$ is the one from my paper [1, Lemma 9]:

```
i16 : L.mukaiPairing
o16 = | 2  -1 0  |
      | -1 2  -1 |
      | 0  -1 -2 |
               3        3
o16 : Matrix ZZ  <--- ZZ
```

The movable cone has two chambers, and geometrically we know that the hypekähler admits a Mukai flop along the Lagrangian $\mathbb{P}^2$ parametrizing lines in the plane in the cubic:

```
i17 : nefChambers(L)
Searching v.w = 0...
Searching v.w = 1...
o17 = {| 1  |, | 3 |, | 1 |}
       | 2  |  | 6 |  | 2 |
       | -1 |  | 1 |  | 1 |
o17 : List
```

We can see that the last wall of the movable cone is isotropic:

```
i18 : for w in nefChambers(L) list L.square(w)
o18 = {8, 40, 0}
```

But it's not in the nef cone:

```
i19 : nefCone(L)
o19 = {| 1  |, | 3 |}
       | 2  |  | 6 |
       | -1 |  | 1 |
o19 : List
```

So the hyperkähler has an Abelian fibration, but only after we flop the $\mathbb{P}^2$.

Lehn, Lehn, Sorger, and van Straten

Let's take the LLSvS eightfold associated to a Pfaffian cubic:

```
i20 : L = LLSvS(14)
o20 : a lattice of K3^[4] type
o20 : K3nLattice
```

The movable cone has three nef chambers, of which the middle one is the LLSvS eightfold:

```
i21 : nefChambers(L)
Searching v.w = 0...
Searching v.w = 1...
Searching v.w = 2...
Searching v.w = 3...
o21 = {| 11 |, | 6  |, | 8 |, | 17 |}
       | 8  |  | 5  |  | 9 |  | 20 |
       | -9 |  | -3 |  | 3 |  | 9  |
o21 : List

i22 : nefCone(L)
o22 = {| 6  |, | 8 |}
       | 5  |  | 9 |
       | -3 |  | 3 |
o22 : List
```

The chamber on the left is Hilb^4 of a degree-14 K3 surface, as is consistent with my paper with Lehn [2], and you can run that Hilbert scheme through the software if you're interested. But the eightfold also has an involution, which we can see as follows:

```
i23 : birationalInvolution(L)
o23 = | 0 1 -1 |
      | 1 0 -1 |
      | 0 0 -1 |
                 3        3
o23 : Matrix ZZ  <--- ZZ
```

This involution reverses the nef chambers:

```
i24 : o23 * o21
o24 = {| 17 |, | 8 |, | 6  |, | 11 |}
       | 20 |  | 9 |  | 5  |  | 8  |
       | 9  |  | 3 |  | -3 |  | -9 |
o24 : List
```

So there are only two birational models, not three.

Double EPW sextics

Let's take double EPW sextics associated to a special Gushel-Mukai fourfold of discriminant 12:

```
i25 : L = doubleEPWsextic 12
o25 : a lattice of K3^[2] type
o25 : K3nLattice
```

The basis for $K_{\mathrm{num}}(\mathcal{A}_X)$ is the one from Pertusi's paper [6, Lemma 4.6]:

```
i26 : L.mukaiPairing
o26 = | 2  0  -1 |
      | 0  2  -1 |
      | -1 -1 -2 |
                3       3
o26 : Matrix ZZ  <--- ZZ
```

Something interesting happens in this example, because the movable cone equals the positive cone, but the walls are irrational:

```
i27 : positiveCone(L)
o27 = {| .591752  |, | 1.40825 |}
       | -.408248 |  | .408248 |
       | -.816497 |  | .816497 |
o27 : List

i28 : movableCone(L)
Searching v.w = 0...
Searching v.w = 1...
o28 = {| .591752  |, | 1.40825 |}
       | -.408248 |  | .408248 |
       | -.816497 |  | .816497 |
o28 : List
```

So the birational automorphism group is infinite, either isomorphic to $\mathbb{Z}$ or to an infinite dihedral group $\mathbb{Z} \rtimes \mathbb{Z}/2$. See Debarre's survey paper [4, Theorems 4.9 and 4.19] for a nice discussion of the generalities here. In particular the movable cone has infinitely many nef chambers, and we certainly can't print them all; instead we do this:

```
i29 : nefChambers(L)
o29 = ({| 5  |, | 7 |, | 25 |, | 59 |, | 245 |}, | 69 0 50 |)
        | -1 |  | 1 |  | 7  |  | 17 |  | 71  |   | 20 1 14 |
        | -2 |  | 2 |  | 14 |  | 34 |  | 142 |   | 40 0 29 |
o29 : Sequence
```

The first item of the output is a list of nef walls. The second item, call it

$$A = \begin{pmatrix} 69 & 0 & 50 \\ 20 & 1 & 14 \\ 40 & 0 & 29 \end{pmatrix},$$

is a generator of that $\mathbb{Z}$ in the birational automorphism group. Notice that A sends the first nef wall to the last one, so every nef chamber differs from one of these four by a power of A. I welcome suggestions for how to make the output more understandable in cases like this.

In this example there's also an involution:

```
i30 : birationalInvolution(L)
o30 = | 1 0 -1 |
      | 0 1 -1 |
      | 0 0 -1 |
                3        3
o30 : Matrix ZZ  <--- ZZ
```

Unlike the LLSvS example, this involution is not unique: multiplying by any power of A gives another one. The involution takes the first chamber to itself, the second chamber to A^{-1} times the last chamber, and so on:

```
i31 : o30 * o29_0
o31 = {| 7 |, | 5  |, | 11  |, | 25  |, | 103  |}
       | 1 |  | -1 |  | -7  |  | -17 |  | -71  |
       | 2 |  | -2 |  | -14 |  | -34 |  | -142 |
o31 : List

i32 : o29_1^-1 * o29_0
o32 = {| 245  |, | 103  |, | 25  |, | 11  |, | 5  |}
       | -169 |  | -71  |  | -17 |  | -7  |  | -1 |
       | -338 |  | -142 |  | -34 |  | -14 |  | -2 |
o32 : List
```

Thus what appeared to be four birational models are in fact only three, and two of them have biregular involutions.

It is interesting to explore the birational isomorphism between this double EPW sextic and the variety of lines on a special cubic 4-fold of discriminant 12.

For discriminants $d \equiv 2 \pmod 8$, there are two families of Gushel–Mukai 4-folds. The syntax for the second family is current as follows:

```
i33 : L = doubleEPWsextic(18, SecondFamily => true)
o33 : a lattice of K3^[2] type
o33 : K3nLattice
```

Build your own

If you want to go beyond the families of varieties discussed above, you can supply your own Mukai pairing and Mukai vector, but the catch is that you have to supply your own ample class too.

For example, suppose you wanted to take a special cubic fourfold of discriminant 20, and the moduli space of complexes Kuznetsov's K3 category with numerical class $2\lambda_1 + 3\lambda_2$:

```
i34 : L = constructK3nLattice(cubicFourfold(20), vector{2,3,0})
o34 : a lattice of K3^[8] type
o34 : K3nLattice
```

But before we can compute any cones, we have to provide an ample class:

```
i35 : setAmpleClass(L, vector{4,-1,0})
```

If you're not sure what to use for an ample class and want to fish around for one, you can use this function:

```
i36 : L.isSafelyAmple(vector{4,-1,0})
o36 = true
```

For a class $h \in v^\perp$ with $h^2 > 0$, this checks that the discriminant of the pairing on $h^\perp$ is greater than $(v^2/2)^2 + 2v^2$, which guarantees that h is not on any nef wall.[1] If you try to set an ample class that doesn't satisfy `isSafelyAmple`, it will complain but it won't stop you. If it later finds that your class is on a nef wall, it will complain again.

Or if you want to take a Gushel–Mukai fourfold of discriminant 10 in the second family, and take the 14-dimensional moduli space of stable complexes with numerical class $\lambda_1 + 2\lambda_2$, you can do this:

```
i37 : L = constructK3nLattice(gushelMukai(10, SecondFamily => true), vector{1,2,0})
o37 : a lattice of K3^[6] type
o37 : K3nLattice

i38 : setAmpleClass(L, vector{2,-1,0})
```

Lastly, you can supply your own Mukai pairing. Suppose you want the Debarre–Voisin fourfold from a special hyperplane section of $\mathrm{Gr}(3, 10)$ of discriminant 24. (Has anyone worked out the Noether–Lefschetz story for these things?)

```
i39 : L = constructK3nLattice(matrix{{2,-1,0},{-1,6,-1},{0,-1,-2}}, vector{1,0,0})
o39 : a lattice of K3^[2] type
o39 : K3nLattice

i40 : setAmpleClass(L, vector{1,2,0})
```

It will raise an error if your Mukai pairing is not symmetric, or not even, or has the wrong signature, or if your Mukai vector is not primitive or its square is not positive.

[1] This is inspired by Debarre and Macrì's new paper [5, Prop. 4.12]. I'll write the proof later.

Acknowledgements

This software was developed in collaboration with Ian McConachie and Leonie Way through the University of Oregon math department's summer undergraduate research program. McConachie was supported by NSF grant no. DMS-2039316. Way was supported by the deparment. I was supported by both of those and by NSF grant no. DMS-1902213.

It is a pleasure to thank Arend Bayer, Emanuele Macrì, and Eyal Markman for patiently answering questions about their work. I also thank Laura Pertusi for advice on Gushel–Mukai fourfolds, Karl Schwede for advice on Macaulay2 programming, Robert Lipshitz for a helpful conversation, and Kevin Buzzard for introducing me to Pell-type equations.

References

[1] N. Addington. On two rationality conjectures for cubic fourfolds. *Math. Res. Lett.*, 23(1):1–13, 2016. Also arXiv:1405.4902.

[2] N. Addington and M. Lehn. On the symplectic eightfold associated to a Pfaffian cubic fourfold. *J. reine angew. Math.*, 731:129–137, 2017. Also arXiv:1404.5657.

[3] A. Bayer and E. Macrì. MMP for moduli of sheaves on K3s via wall-crossing: nef and movable cones, Lagrangian fibrations. *Invent. Math.*, 198(3):505–590, 2014. Also arXiv:1301.6968.

[4] O. Debarre. Hyperkähler manifolds. Preprint, arXiv:1810.02087.

[5] O. Debarre and E. Macrì. Complete curves in the moduli space of polarized K3 surfaces and hyper-Kähler manifolds.

[6] L. Pertusi. On the double EPW sextic associated to a Gushel-Mukai fourfold. *J. Lond. Math. Soc. (2)*, 100(1):83–106, 2019. Also arXiv:1709.02144.